Q:

refresh pivot table in multiple active workbooks at once

I need to routinely refresh multiple pivot tables from different workbook from our SQL database. 
What's the easiest way to do this?  Refreshing pivot table in different workbook at once?  its alright if it is in vba format. 

A:

ok .below code will ask you the folder path where the excel files exist(which needs to be refreshed.) 
Source : 
Sub LoopAllExcelFilesInFolder()

'PURPOSE: To loop through all Excel files in a user specified folder and perform a set task on them
'SOURCE: www.TheSpreadsheetGuru.com

Dim wb As Workbook
Dim myPath As String
Dim myFile As String
Dim myExtension As String
Dim FldrPicker As FileDialog

'Optimize Macro Speed
  Application. ScreenUpdating = False
  Application. EnableEvents = False
  Application. Calculation = xlCalculationManual

'Retrieve Target Folder Path From User
  Set FldrPicker = Application. FileDialog(msoFileDialogFolderPicker)

    With FldrPicker
      .Title = "Select A Target Folder"
      .AllowMultiSelect = False
        If .Show <> -1 Then GoTo NextCode
        myPath = .SelectedItems(1) & "\"
    End With

'In Case of Cancel
NextCode:
  myPath = myPath
  If myPath = "" Then Exit Sub

'Target File Extension (must include wildcard "*")
  myExtension = "*.xls"

'Target Path with Ending Extention
  myFile = Dir(myPath & myExtension)

'Loop through each Excel file in folder
  Do While myFile <> ""
    'Set variable equal to opened workbook
      Set wb = Workbooks. Open(Filename:=myPath & myFile)

    'Refresh the workbook
      wb.refreshall

    'Save and Close Workbook
      wb. Close SaveChanges:=True

    'Get next file name
      myFile = Dir
  Loop

'Message Box when tasks are completed
  MsgBox "Task Complete!" 

'Reset Macro Optimization Settings
  Application. ScreenUpdating = True
  Application. EnableEvents = True
  Application. Calculation = xlCalculationAutomatic

End Sub

